# NAG Fortran Library Routine Document

# G05MJF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

## 1 Purpose

G05MJF generates a vector of pseudo-random integers from the discrete binomial distribution with parameters $m$ and $p$.

## 2 Specification

```
SUBROUTINE G05MJF(MODE, M, P, N, X, IGEN, ISEED, R, NR, IFAIL)
INTEGER           MODE, M, N, X(N), IGEN, ISEED(4), NR, IFAIL
real              P, R(NR)
```

## 3 Description

G05MJF generates $n$ integers $x_i$ from a discrete binomial distribution, where the probability of $x_i = I$ is

$$P(x_i = I) = \frac{m!}{I!(m-I)!} p^I \times (1-p)^{m-I}, \quad I = 0, 1, \ldots, m,$$

where $0 \leq m$ and $0 \leq p \leq 1$. This represents the probability of achieving $I$ successes in $m$ trials when the probability of success at a single trial is $p$.

The variates can be generated with or without using a search table and index. If a search table is used then it is stored with the index in a reference vector and subsequent calls to G05MJF with the same parameter values can then use this reference vector to generate further variates.

One of the initialisation routines G05KBF (for a repeatable sequence if computed sequentially) or G05KCF (for a non-repeatable sequence) must be called prior to the first call to G05MJF.

## 4 References

Knuth D E (1981) *The Art of Computer Programming (Volume 2)* (2nd Edition) Addison-Wesley

Kendall M G and Stuart A (1969) *The Advanced Theory of Statistics (Volume 1)* (3rd Edition) Griffin

## 5 Parameters

1: MODE – INTEGER *Input*

*On entry*: a code for selecting the operation to be performed by the routine:

MODE $= 0$

Set up reference vector only.

MODE $= 1$

Generate variates using reference vector set up in a prior call to G05MJF.

MODE $= 2$

Set up reference vector and generate variates.

MODE $= 3$

Generate variates without using the reference vector.

*Constraint*: $0 \leq$ MODE $\leq 3$.

2: M – INTEGER *Input*

*On entry*: the number of trials, $m$, of the distribution.

*Constraint*: $\mathrm{M} \geq 0$.

3: P – ***real*** *Input*

*On entry*: the probability of success $p$ of the binomial distribution.

*Constraint*: $0.0 \leq \mathrm{P} \leq 1.0$.

4: N – INTEGER *Input*

*On entry*: the number, $n$, of pseudo-random numbers to be generated.

*Constraint*: $\mathrm{N} \geq 1$.

5: X(N) – INTEGER array *Output*

*On exit*: the $n$ pseudo-random numbers from the specified binomial distribution.

6: IGEN – INTEGER *Input*

*On entry*: must contain the identification number for the generator to be used to return a pseudo-random number and should remain unchanged following initialisation by a prior call to one of the routines G05KBF or G05KCF.

7: ISEED(4) – INTEGER array *Input/Output*

*On entry*: contains values which define the current state of the selected generator.

*On exit*: contains updated values defining the new state of the selected generator.

8: R(NR) – ***real*** array *Input/Output*

*On exit*: the reference vector.

9: NR – INTEGER *Input*

*On entry*: the dimension of the array R as declared in the (sub)program from which G05MJF is called.

*Suggested value*: $\mathrm{NR} = 22 + 20\sqrt{\mathrm{M} \times \mathrm{P}(1-\mathrm{P})}$.

*Constraints*:

if MODE $= 0$ or $2$, then

$$\begin{aligned}\mathrm{NR} > \quad & \min(\mathrm{M}, \mathrm{INT}[\mathrm{M} \times \mathrm{P} + 7.15\sqrt{\mathrm{M} \times \mathrm{P}(1-\mathrm{P})} + 1]) \\ & - \max(0, \mathrm{INT}[\mathrm{M} \times \mathrm{P} - 7.15\sqrt{\mathrm{M} \times \mathrm{P}(1-\mathrm{P})} - 7.15]) + 6;\end{aligned}$$

if MODE $= 1$, then NR should remain unchanged from the previous call to G05MJF;
if MODE $= 3$, then R is not referenced.

10: IFAIL – INTEGER *Input/Output*

*On entry*: IFAIL must be set to $0$, $-1$ or $1$. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

*On exit*: IFAIL $= 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value $-1$ or $1$ is recommended. If the output of error messages is undesirable, then the value $1$ is recommended. Otherwise, for users not familiar with this parameter the recommended value is $0$. **When the value $-1$ or $1$ is used it is essential to test the value of IFAIL on exit.**

## 6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or $-1$, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, N $< 1$.

IFAIL $= 2$

On entry, NR is too small when MODE $= 0$ or 2 (see Section 5).

IFAIL $= 3$

On entry, P $< 0.0$
or P $> 1.0$.

IFAIL $= 4$

On entry, M $< 0$.

IFAIL $= 5$

On entry, MODE $< 0$
or MODE $> 3$.

IFAIL $= 6$

M or P is not the same as when R was set up in a previous call with MODE $= 0$ or 2.

## 7 Accuracy

Not applicable.

## 8 Further Comments

None.

## 9 Example

The example program prints 20 pseudo-random integers from a binomial distribution with parameters $m = 6000$ and $p = 0.8$, generated by a single call to G05MJF, after initialisation by G05KBF.

### 9.1 Program Text

**Note:** the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     G05MJF Example Program Text
*     Mark 20 Release. NAG Copyright 2001.
*     .. Parameters ..
      INTEGER          NOUT, N, NR
      PARAMETER        (NOUT=6,N=20,NR=6007)
*     .. Local Scalars ..
      real             P
      INTEGER          I, IFAIL, IGEN, M
*     .. Local Arrays ..
      real             R(NR)
      INTEGER          ISEED(4), X(N)
*     .. External Subroutines ..
      EXTERNAL         G05KBF, G05MJF
```

```
*     .. Executable Statements ..
      WRITE (NOUT,*) 'G05MJF Example Program Results'
      WRITE (NOUT,*)
*     Set the distribution parameters P and M
      P = 0.8e0
      M = 6000
*     Initialise the seed to a repeatable sequence
      ISEED(1) = 1762543
      ISEED(2) = 9324783
      ISEED(3) = 42344
      ISEED(4) = 742355
*     IGEN identifies the stream.
      IGEN = 1
      CALL G05KBF(IGEN,ISEED)
*     Choose MODE = 2
      IFAIL = 0
      CALL G05MJF(2,M,P,N,X,IGEN,ISEED,R,NR,IFAIL)
*
      WRITE (NOUT,99999) (X(I),I=1,N)
      STOP
*
99999 FORMAT (1X,I12)
      END
```

## 9.2 Program Data

None.

## 9.3 Program Results

```
 G05MJF Example Program Results

         4758
         4851
         4793
         4820
         4851
         4795
         4807
         4792
         4787
         4842
         4801
         4794
         4806
         4878
         4745
         4790
         4832
         4789
         4743
         4812
```